

Using a Dynamic Panel Estimator to Model Change in Panel Data

Roberto G. Gutierrez, SAS Institute Inc.
A. Rachid El-Khattabi, University of North Carolina at Chapel Hill

ABSTRACT

Panel data, which are collected on a set (panel) of individuals over several time points, are ubiquitous in economics and other analytic fields because their structure allows for individuals to act as their own control groups. The PANEL procedure in SAS/ETS[®] software models panel data that have a continuous response, and it provides many options for estimating regression coefficients and their standard errors. Some of the available estimation methods enable you to estimate a dynamic model by using a lagged dependent variable as a regressor, thus capturing the autoregressive nature of the underlying process. Including lagged dependent variables introduces correlation between the regressors and the residual error, which necessitates using instrumental variables. This paper guides you through the process of using the typical estimation method for this situation—the generalized method of moments (GMM)—and the process of selecting the optimal set of instrumental variables for your model. Your goal is to achieve unbiased, consistent, and efficient parameter estimates that best represent the dynamic nature of the model.

INTRODUCTION

Panel data occur when you observe a panel of individuals—such as people, households, and corporations—over a period of time during which you obtain several observations per individual. Panel data have two dimensions: the individual dimension (or *cross section*) and the time dimension. For example, suppose you have data from a panel of 30 firms, and for each firm you record net sales in each year from 1990 through 1999. Your panel data consist of 300 observations, but they provide more value than 300 observations of random firms in random years. Because you follow the same 30 firms over the same time frame, it is less likely that model parameters are confounded by firm effects, time effects, or both. Put simply, each firm (and each time period) acts as its own control group.

Dynamic panel models are linear regression models that are generalized in two ways:

1. They include individual effects, yielding a two-tiered error structure: individual-level errors and overall residual errors.
2. They allow the dependent variable to depend on its value from the previous time period, thus making the model *dynamic*.

The first generalization adds individual (cross-sectional) effects to linear regression. Formally, for individual $i = 1, \dots, N$ at time $t = 1, \dots, T_i$,

$$y_{it} = \beta_0 + \beta_x \mathbf{X}_{it} + v_i + \epsilon_{it}$$

The model contains a set of explanatory variables \mathbf{X}_{it} . The v_i are individual effects, and the ϵ_{it} are observation-level regression errors. In SAS/ETS 14.2, the PANEL procedure supports 15 different ways to fit this model, depending on the properties of the two-tiered error structure and how that structure interacts with \mathbf{X}_{it} . At opposite ends of the spectrum are the generalized least squares (GLS) random-effects estimator and the fixed-effects estimator, which make the strictest and least strict assumptions, respectively, about the individual effects.

The second generalization of linear regression results in an autoregressive (AR) model. Ignoring individual effects for now, consider a regression model that includes a “lagged” version of the dependent variable:

$$y_{it} = \beta_0 + \phi y_{i, t-1} + \beta_x \mathbf{X}_{it} + \epsilon_{it}$$

The model is *dynamic* because the equation for time t includes an element from the previous time period, the lagged response $y_{i, t-1}$.

In the dynamic AR model, the dependent variable depends on its value from the previous time period in a way that is not explained by the regressors \mathbf{X}_{it} . One way to interpret ϕ is that when $|\phi| < 1$, ϕ is the correlation between y_{it} and $y_{i, t-1}$.

Alternatively, ϕ is the degree to which change (or shock) affects the system. If you are tallying net sales for a company, suppose that a policy change in 2002—a policy change not represented by \mathbf{X}_{it} —resulted in much higher sales for that year. The higher sales are represented in the AR model by a large positive residual ϵ for the year 2002. You do not expect the effect of the policy change to instantaneously vanish on January 1, 2003. Therefore, you include lagged sales in your model so that sales for 2003 depend on those for 2002. This way, year 2003 sales also benefit from the 2002 policy change. Moving forward, sales for 2004 depend on those from 2003, and by construction also depend on 2002 and the policy change. Over the years, the effect of the policy change diminishes if $|\phi| < 1$. You interpret ϕ as the degree to which the shock carries through the system. If $\phi = 0.9$, then the effect diminishes gradually. If $\phi = 0.1$, then things go back to normal fairly quickly.

You can fit AR models by using the AUTOREG procedure in SAS/ETS. As a technical note, AR models work best when $|\phi| < 1$. Otherwise, the data are *nonstationary* and estimation results require different interpretation. Nonstationary time series data, and the problems they present, are beyond the scope of this paper.

You obtain a *dynamic panel model* by adding cross-sectional effects (v_i) to the AR model:

$$y_{it} = \beta_0 + \phi y_{i, t-1} + \beta_x \mathbf{X}_{it} + v_i + \epsilon_{it}$$

Dynamic panel models are a simple concept, but the devil is in the details. It is easy to form dynamic panel models and interpret estimation results. However, obtaining the estimation results requires specialized methodology and some fine-tuning on your part. Having a lagged dependent variable in the right-hand side of the model introduces correlation that biases standard panel-data estimators. You can alleviate the bias by using *instrumental variables*, which are variables that are not correlated with the regression errors. Choosing an optimal set of instrumental variables requires you to apply substantive knowledge of how the variables in your data interact with each other. This paper will get you started on the process of using the PANEL procedure in SAS/ETS to fit dynamic panel models, beginning with standard linear regression and proceeding from there.

Dynamic panel analysis is an important subfield of econometrics, with most of the research dedicated to determining the optimal estimation strategy given a set of data conditions. Baltagi (2008) provides a chapter that introduces dynamic-panel estimation, and Wooldridge (2010) covers the fundamentals of estimating dynamic panel (and similar) models.

DATA ON BEEF SALES

Consider hypothetical sales data for 48 states, for which per capita beef sales and beef prices were recorded each year from 2002 through 2011. The following SAS statements create the data set Beef:

```
data Beef;
  input State Year Sales Price;
  label Sales = "Log of per capita beef sales";
  label Price = "Log of beef price";
  datalines;
1 2002 5.05050 1.15128
1 2003 6.39724 1.54282
1 2004 6.12989 1.55918
1 2005 6.52674 1.51257
1 2006 6.61818 1.74706
1 2007 6.95997 1.92716
1 2008 7.29500 1.82181
1 2009 6.45306 1.90273
1 2010 6.25180 2.00069
1 2011 6.21774 2.26738
2 2002 3.79823 1.28592
2 2003 4.83639 1.46810
2 2004 5.33434 1.43959
2 2005 5.23270 1.36039

... more lines ...
```

The variables **State** and **Year** are identification variables that represent state and year, respectively. The variable **Sales** records the logarithm of per capita beef sales (in pounds), and the variable **Price** is the logarithm of the average beef price per pound.

You want to determine the association between beef prices and sales. You suspect that sales differ by state in ways other than by price and, because your data are limited, you suspect that there are unmeasured factors that might influence both present and future sales. Therefore, you posit a dynamic panel model that includes state effects and a lagged dependent variable. For state i in year t ,

$$\text{Sales}_{it} = \beta_0 + \phi \text{Sales}_{i,t-1} + \beta_1 \text{Price}_{it} + v_i + \epsilon_{it}$$

Your first order of business is to produce a version of your data that contains first-order lags of sales. The following SAS statements demonstrate how to use the PANEL procedure to generate lagged variables:

```
proc panel data = Beef;
  id State Year;
  clag Sales(1) / out = BeefLagged;
run;
```

In the preceding code, the ID statement specifies the state and time variables, in that order. The code does not perform any estimation, but produces the data set BeefLagged, which is a copy of the original data plus the added variable **Sales_1**, which contains the lagged values of **Sales**.

The PANEL procedure provides great flexibility in generating lagged variables. You can generate lags for one or more variables, you can generate higher-order lags, and you can specify how to handle missing values that result when no prior time period exists. For example, the CLAG statement replaces missing values with cross-sectional means.

For more information about generating lagged variables, see the chapter about the PANEL procedure in the *SAS/ETS 14.2 User's Guide*.

POOLED REGRESSION

As a starting point, you fit the dynamic panel model to your data by ordinary least squares (OLS). In panel-data parlance, this is known as *pooled regression* because you pool all the data together without regard to state affiliation.

To fit the model by using the PANEL procedure, use a MODEL statement that includes lagged sales as a right-hand side (RHS) variable, and specify the POOLED option:

```
proc panel data = BeefLagged;
  id State Year;
  model Sales = Sales_1 Price / pooled;
run;
```

Figure 1 shows the results. The output provides descriptive information about the estimation method and the data, regression fit statistics such as mean-squared error (MSE) and R^2 , and a table of parameter estimates.

Figure 1 Beef Sales Data, Pooled Estimation

The PANEL Procedure Pooled (OLS) Estimates

Dependent Variable: Sales (Log of per capita beef sales)

Model Description			
Estimation Method		Pooled	
Number of Cross Sections		48	
Time Series Length		10	
Fit Statistics			
SSE	175.3621	DFE	477
MSE	0.3676	Root MSE	0.6063
R-Square	0.2605		

Figure 1 continued

Parameter Estimates						
Variable	DF	Estimate	Standard Error	t Value	Pr > t	Label
Intercept	1	2.651903	0.2465	10.76	<.0001	Intercept
Sales_1	1	0.405989	0.0423	9.60	<.0001	Lagged per capita beef sales
Price	1	0.466423	0.0977	4.77	<.0001	Log of beef price

Because pooled estimation does not consider state affiliation, the estimates do not account for the state effects ν_i . Ignoring that deficiency for now, you interpret the coefficient on **Sales_1** (0.406) as the correlation between sales for the current year and sales for the previous year, after controlling for price. The autocorrelation is statistically significant, indicating that you have unmeasured factors affecting sales and that the effects of these factors linger for several years.

Because your data consist of the logarithms of both sales and price, you interpret the coefficient for **Price** (0.466) as the demand elasticity to price. For example, a 10% increase in price would result in a 4.66% increase in sales. However, that coefficient is probably inaccurate because you omitted state effects—effects that can be shown to be correlated with lagged sales—from the analysis.

INCLUDING CROSS-SECTIONAL EFFECTS

As stated in the previous section, pooled estimation ignores cross-sectional effects. To incorporate these effects in a way that makes minimal assumptions, you can use a method that “sweeps” (removes) the cross-sectional effects out of the estimation. Removing the effects also means that you do not have to estimate them directly, which is convenient for large data sets.

Two transformations that remove cross-sectional effects from the estimation are the *within* transformation and the *first-differencing* transformation.

THE WITHIN TRANSFORMATION

The within transformation subtracts from each observation its cross-sectional mean. Applying this transformation to both sides of the dynamic panel model yields

$$y_{it} - \bar{y}_i = \phi(y_{i, t-1} - \bar{y}_{i, -1}) + \beta_x(\mathbf{X}_{it} - \bar{\mathbf{X}}_i) + (\epsilon_{it} - \bar{\epsilon}_i)$$

where the subscript $i, -1$ denotes an average of all observations for individual i except the first (for which no lagged value exists). Because any effect ν_i is common to all observations for individual i , the effect is removed by the within transformation.

To apply this estimation method to your data, use the `FIXONE` option in the `MODEL` statement as follows:

```
proc panel data = BeefLagged;
  id State Year;
  model Sales = Sales_1 Price / noint fixone;
run;
```

The `NOINT` option suppresses estimation of the intercept term. Suppressing the intercept term has no effect on the estimation because the within transformation also sweeps out β_0 .

Figure 2 provides the results. In addition to the results you would find with pooled estimation, the `PANEL` procedure provides an F test to determine whether state effects are statistically significant. The null hypothesis is that the state effects are all 0, and based on the output you reject that hypothesis.

Figure 2 Beef Sales Data, Estimation by the Within Transformation

**The PANEL Procedure
Fixed One-Way Estimates**

Dependent Variable: Sales (Log of per capita beef sales)

Model Description			
Estimation Method		FixOne	
Number of Cross Sections		48	
Time Series Length		10	
Fit Statistics			
SSE	134.8807	DFE	430
MSE	0.3137	Root MSE	0.5601
R-Square	0.4312		
F Test for No Fixed Effects and No Intercept			
Num DF	Den DF	F Value	Pr > F
48	430	5.52	<.0001

Parameter Estimates						
Variable	DF	Estimate	Standard Error	t Value	Pr > t	Label
Sales_1	1	0.070878	0.0490	1.45	0.1489	Lagged per capita beef sales
Price	1	0.74586	0.0939	7.95	<.0001	Log of beef price

THE FIRST-DIFFERENCING TRANSFORMATION

The first-differencing transformation subtracts the previous observation from the current observation:

$$y_{it} - y_{i, t-1} = \phi(y_{i, t-1} - y_{i, t-2}) + \beta_x(\mathbf{X}_{it} - \mathbf{X}_{i, t-1}) + (\epsilon_{it} - \epsilon_{i, t-1})$$

When you perform this transformation on all individuals and at all time points, you form the set of *difference equations*.

You obtain first-differencing estimates by specifying the FDONE option in the MODEL statement:

```
proc panel data = BeefLagged;
  id State Year;
  model Sales = Sales_1 Price / noint fdone;
run;
```

Figure 3 provides the parameter estimates table, which shows that the estimates are markedly different from those produced by the within transformation.

Figure 3 Beef Sales Data, Estimation by First Differencing

**The PANEL Procedure
First Difference Estimates for One-Way**

Dependent Variable: Sales (Log of per capita beef sales)

Parameter Estimates						
Variable	DF	Estimate	Standard Error	t Value	Pr > t	Label
Sales_1	0	-0.33234	0.0331	-10.03	<.0001	Lagged per capita beef sales
Price	0	0.368672	0.1453	2.54	0.0115	Log of beef price

ENDOGENOUS-REGRESSOR BIAS

A standard assumption of linear regression is that the residual errors are independent of the regressors. Regressors that violate this assumption are said to be *endogenous*. When none of the regressors are endogenous, first differencing should produce results similar to those produced by the within transformation.

Transforming the model redefines the residual. In the within transformation, the residual is $(\epsilon_{it} - \bar{\epsilon}_i)$; in the first-differencing transformation, the residual is $(\epsilon_{it} - \epsilon_{i, t-1})$. In both transformed models, the regression variable that contains the lagged response is an endogenous regressor. Because the transformations differ, the details of the endogeneity also differ. Therefore, the parameter estimates differ according to which transformation you use.

In the within transformation, Nickell (1981) noted that $(\epsilon_{it} - \bar{\epsilon}_i)$ is correlated with $(y_{i, t-1} - \bar{y}_i, -1)$ because of the following conditions:

- $\bar{\epsilon}_i$ contains $\epsilon_{i, t-1}$, which is the error term that helps form $y_{i, t-1}$.
- $\bar{y}_i, -1$ contains y_{it} , which is formed by its error term ϵ_{it} .

In the first-differencing transformation, $(\epsilon_{it} - \epsilon_{i, t-1})$ is correlated with $(y_{i, t-1} - y_{i, t-2})$ because $\epsilon_{i, t-1}$ is a component of $y_{i, t-1}$.

It follows that both the within and the first-differencing estimators are biased and require modification.

SIMPLE DYNAMIC PANEL ESTIMATION

One way to alleviate the bias caused by endogenous regressors is to use *instrumental variables*—variables that can predict the endogenous regressor but that themselves are not endogenous (that is, the variables are *exogenous*). The motivation is as follows: if you have an endogenous variable z , then take z out of the model and substitute in its place a prediction, \hat{z} , that is obtained from exogenous variables. That way, z plays a role in the model only through its dependence on exogenous variables—no assumption is violated. All that remains is how to choose effective instrumental variables for the data at hand.

In dynamic panels, the problem of choosing appropriate instrumental variables is made easier because you know the source of the endogeneity: you included a lagged dependent variable in the model.

It is simpler to work with the first-differencing estimator and alleviate the correlation between $(\epsilon_{it} - \epsilon_{i, t-1})$ and $(y_{i, t-1} - y_{i, t-2})$. The cause of that correlation is the correlation between $y_{i, t-1}$ and $\epsilon_{i, t-1}$. Thus, you require instrumental variables that can predict $y_{i, t-1}$ but are independent of $\epsilon_{i, t-1}$.

This is where the autocorrelation structure of your model works in your favor. You can predict $y_{i, t-1}$ from its previous value $y_{i, t-2}$, and $y_{i, t-2}$ is independent of $\epsilon_{i, t-1}$ because $y_{i, t-2}$ occurred *before* $\epsilon_{i, t-1}$. Begin with $t = 3$, the first time period for which the difference equations hold. You can use y_{i1} to predict y_{i2} , y_{i2} to predict y_{i3} , and so on.

The following SAS statements use the PANEL procedure to implement this estimation:

```
proc panel data = BeefLagged;
  id State Year;
  instruments depvar;
  model Sales = Sales_1 Price / gmm2 noint nolevels robust artest = 2;
run;
```

The syntax differs from that in the previous section as follows:

- The INSTRUMENTS statement specifies which variables to use as instruments. You specify the DEPVAR option to use instruments that are lagged versions of the dependent variable.
- The GMM2 option specifies estimation by the two-step generalized method of moments (GMM). Also available are one-step GMM (the GMM1 option) and iterative GMM (the ITGMM option). Two-step GMM is more popular because it is less sensitive to model misspecification than one-step GMM. Also, two-step GMM does not need to converge to a solution; iterative GMM requires convergence.
- The NOLEVELS option specifies that no level equations are to be used in the estimation and that only difference equations are to be used. The resulting estimator is a direct extension of the FDONE estimator from the previous section.

- The ROBUST option specifies robust standard errors. With two-step GMM, robust standard errors are more reliable than conventional model-based standard errors.
- The ARTEST= option specifies the maximum order for performing diagnostic tests of serial correlation in the residuals.

Figure 4 Beef Sales Data, Simple Dynamic Panel Estimation

**The PANEL Procedure
GMM: First Differences Transformation**

Dependent Variable: Sales (Log of per capita beef sales)

Model Description				
Estimation Method	GMM2			
Number of Cross Sections	48			
Time Series Length	10			
Estimate Stage	2			
Maximum Number of Time Periods (MAXBAND)	1			

Fit Statistics			
SSE	8056.3517	DFE	382
MSE	21.0899	Root MSE	4.5924

Sargan Test		
DF	Statistic	Prob > ChiSq
6	360.52	<.0001

Parameter Estimates					
	DF	Estimate	Standard Error	t Value	Pr > t
Sales_1	1	0.353772	0.00360	98.21	<.0001
Price	1	-0.35524	0.00939	-37.82	<.0001

AR(m) Test		
Lag	Statistic	Pr > Statistic
1	-5.16	2.456E-7
2	1.02	0.3058

Figure 4 provides the results. Here is the guided tour:

- The “Model Description” and “Fit Statistics” tables provide general model information and summary fit statistics; they are similar to what you would find with other estimation methods.
- The “Sargan Test” table provides a test of overidentifying restrictions—that is, a test of the null hypothesis that the instrument set is appropriate for the data at hand. Because the p -value is less than 0.0001, you reject that null hypothesis. Typically, rejecting the null hypothesis would indicate that there is a problem with the instrument set. However, Sargan tests are known to overreject when you have significant individual effects. Because state effects are evident from the data, the small p -value is probably not cause for alarm.
- The “Parameter Estimates” table shows estimates that are very different from those produced by estimation via first differencing (FDONE). This difference demonstrates the advantage of dynamic panel estimation. The degree of autocorrelation is moderate but statistically significant. Price elasticity is now negative. Increases in price are associated with drops in sales.
- The “AR(m) Test” table provide tests for serial correlation in the residuals. Because you specified ARTEST=2 in the MODEL statement, you obtain AR tests for lag orders 1 and 2.

The theory behind dynamic panel estimation indicates that you should see some lag 1 serial correlation in model residuals but that residuals should be uncorrelated when the lag is 2 or greater. The results agree with those expectations.

ARELLANO-BOND ESTIMATION

Rather than predict the lagged dependent variable based solely on its previous value, Arellano and Bond (1991) propose a larger set of instruments that use *all* previous values to predict the present. Under this scenario, you use y_{i1} to predict y_{i2} ; y_{i1} and y_{i2} to predict y_{i3} ; y_{i1} , y_{i2} , and y_{i3} to predict y_{i4} ; and so on.

To implement the Arellano-Bond estimator, use the MAXBAND= option to set the maximum bandwidth to $T - 2$, which equals 8 for these data because you have 10 observations per state:

```
proc panel data = BeefLagged;
  id State Year;
  instruments depvar;
  model Sales = Sales_1 Price / gmm2 noit nolevels robust artest = 2
                                maxband = 8;
run;
```

Figure 5 provides the results. The estimates do not differ substantially from the previous section. However, the standard errors are smaller, indicating that the Arellano-Bond set of instruments is more efficient for these data.

Figure 5 Beef Sales Data, Arellano-Bond Estimation

The PANEL Procedure GMM: First Differences Transformation

Dependent Variable: Sales (Log of per capita beef sales)

Model Description			
Estimation Method	GMM2		
Number of Cross Sections	48		
Time Series Length	10		
Estimate Stage	2		
Maximum Number of Time Periods (MAXBAND)	8		

Fit Statistics			
SSE	8286.6939	DFE	382
MSE	21.6929	Root MSE	4.6576

Sargan Test			
DF	Statistic	Prob > ChiSq	
34	1471.90	<.0001	

Parameter Estimates					
Variable	DF	Estimate	Standard Error	t Value	Pr > t
Sales_1	1	0.344679	0.00220	156.91	<.0001
Price	1	-0.36096	0.00474	-76.16	<.0001

AR(m) Test		
Lag	Statistic	Pr > Statistic
1	-5.02	5.242E-7
2	1.01	0.3126

SPECIFYING OTHER INSTRUMENTS

The PANEL procedure is quite flexible in specifying the instruments in dynamic panel estimation. In addition to lags of the dependent variable, you can include past, present, and future values of other variables as instruments. For example, you could include the values of **Price** as instruments in the previous estimation.

The INSTRUMENTS statement makes this easy because it determines the appropriate lags to use and when to use them, based on the properties you set forth for these variables. In addition to specifying the DEPVAR option (to use instruments that are lagged versions of the dependent variable), you can specify the following variable lists:

1. EXOGENOUS=(*varlist*) specifies variables that are uncorrelated with the past, present, and future errors, ϵ_{it} .
2. PREDETERMINED=(*varlist*) specifies variables whose current and past values are uncorrelated with the present error, but whose future values might be correlated with the present error.
2. CORRELATED=(*varlist*) specifies variables that are correlated with the individual effects ν_i , but otherwise are either exogenous or predetermined.

For more information about specifying instruments, see the section “INSTRUMENTS Statement” in the chapter about the PANEL procedure in the *SAS/ETS 14.2 User's Guide*.

SYSTEM GENERALIZED METHOD OF MOMENTS (SYSTEM GMM)

Blundell and Bond (1998) suggest a dynamic panel estimator that jointly estimates the difference equations and the original dynamic-panel equations, where the latter are called *level equations*. Although such duplication seems redundant, you can gain efficiency because what you assume about the regression variables (and their lags and differences) produces distinct sets of restrictions, called *moment conditions*, over both sets of equations. The different moment conditions then lead to different instruments.

When used with an INSTRUMENTS statement that categorizes regressors as exogenous, predetermined, or correlated, the PANEL procedure determines which levels, lags, and differences of the variables to use in the difference equations and which to use in the level equations.

System GMM is the default dynamic panel estimator in the PANEL procedure. Because the previous examples used the NOLEVELS option, the level equations were suppressed and estimation was based solely on the difference equations. You obtain system GMM estimates by simply leaving off the NOLEVELS option.

For more information about system GMM, see the section “Details: PANEL Procedure” in the chapter about the PANEL procedure in the *SAS/ETS 14.2 User's Guide*.

SUMMARY

Dynamic panel models are regression models that include both cross-sectional effects and a lagged dependent variable. They are a powerful tool for dealing with omitted factors whose effects persist over time. Because a lagged dependent variable is included as a regressor, standard panel estimators such as the within estimator are biased. Estimation that is based on the generalized method of moments (GMM) alleviates the bias, provided you choose an appropriate set of instrumental variables. Usually, the optimal set of instrumental variables includes lagged values of the dependent variable. You can base your estimates on either difference equations or level equations, or you can use a system GMM estimator that contains both. The PANEL procedure in SAS/ETS 14.2 supports various dynamic-panel estimation methods, options, and diagnostic tests to help you obtain consistent and efficient estimates that best represent the dynamic nature of the model.

REFERENCES

- Arellano, M., and Bond, S. (1991). “Some Tests of Specification for Panel Data: Monte Carlo Evidence and an Application to Employment Equations.” *Review of Economic Studies* 58:277–297.
- Baltagi, B. H. (2008). *Econometric Analysis of Panel Data*. 4th ed. Chichester, UK: John Wiley & Sons.

- Blundell, R., and Bond, S. (1998). "Initial Conditions and Moment Restrictions in Dynamic Panel Data Models." *Journal of Econometrics* 87:115–143.
- Nickell, S. J. (1981). "Biases in Dynamic Models with Fixed Effects." *Econometrica* 49:1417–1426.
- Wooldridge, J. M. (2010). *Econometric Analysis of Cross Section and Panel Data*. 2nd ed. Cambridge, MA: MIT Press.

ACKNOWLEDGMENT

The supporting work for this paper was completed by Mr. El-Khattabi while he was working as a student scholar at SAS.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author:

Roberto G. Gutierrez
SAS Institute Inc.
SAS Campus Drive
Cary, NC 27513
Bobby.Gutierrez@sas.com

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration.

Other brand and product names are trademarks of their respective companies.